

# Rapport d'étape

Jean-René Licois

## Résumé

Ce document résume un travail en commun avec Guy Barles, Christine Georgelin et Philippe Lescoat dans le cadre du projet MODINGRE en collaboration avec le ppf CASCIMODOT et les départements PHASE, CEPIA et ALIMH de l'INRA. Il s'agit de modéliser le transport et la dégradation des nutriments tout au long de l'intestin grêle chez les mammifères monogastriques.

## 1 Introduction

L'objectif est de modéliser le transport et la digestion des nutriments dans l'intestin grêle. Les hypothèses communes à tous nos modèles sont actuellement :

- ◆ L'intestin grêle peut être considéré comme unidimensionnel : pas de prise en compte de la mécanique des fluides.
- ◆ Les nutriments sont d'une seule espèce.
- ◆ Les nutriments qui sortent de l'estomac sont fractionnés en « paquets » qui contiennent aussi les enzymes endogènes produits par le tube digestif.
- ◆ Le transport de ces « paquets » est assuré par les contractions des diverses couches musculaires situées en périphérie de l'intestin grêle.
  - ✗ L'efficacité du transport est proportionnelle à la taille du « paquet » : plus il y a de matière plus le transport est rapide ...
  - ✗ L'efficacité du transport décroît avec la distance : plus on s'éloigne du duodénum moins le « paquet » avance vite.
- ◆ La dégradation et l'absorption sont dus à une hydrolyse enzymatique. L'activité des enzymes
  - ✗ dépend du  $pH$  de l'intestin au point considéré. Celui-ci est fonction de la distance de ce point au duodénum,
  - ✗ diminue au cours du temps.

Notre première modélisation utilisait une équation aux dérivées partielles de type *équation de transport* qui fut vite remplacé par un modèle utilisant des équations différentielles ordinaires couplées. Les paquets sont assimilés à des sphères de contenu homogène et avancent grâce à des impulsions. Ce modèle s'est enrichi progressivement

- ◆ en considérant l'évolution de plusieurs « paquets »
- ◆ en considérant que l'absorption se fait en périphérie du « paquet »
- ◆ en tenant compte des sécrétions pancréatiques,
- ◆ en remplaçant les « paquets » sphériques par des « paquets » cylindriques,
- ◆ en considérant que les nutriments sont constitués de deux ou trois espèces dont les concentrations sont liées par une loi de comportement.

## 2 Equation de Transport

Notre tout premier modèle utilise une équation aux dérivées partielles de type *équation de transport*

$$\frac{dy}{dt} = -c \frac{dy}{dx}$$

où  $c(x, t)$  est la vitesse de transport.

L'intestion grêle est représenté par l'intervalle  $]0, L[$  (ici  $L = 18$  mètres); un flux de nutriments est injecté en O. On notera  $A(x, t)$  la concentration en nutriment à l'instant  $t$  et à la position  $x$ . Ces nutriments sont digérés grâce à une enzyme de concentration  $z(x, t)$  et on a la réaction  $A + z \mapsto B + z$  où  $B(x, t)$  est la concentration en aliments digérés, donc assimilables par l'organisme.

Le  $pH$  dans l'intestin  $pH$  sera une fonction donnée, typiquement  $ph(x) = 5 + \frac{5x}{16}$ . L'efficacité des enzymes dépend du  $pH$  : dans un premier temps on a envisagé une fonction d'efficacité  $k$  de type gaussienne centrée en 5,5.

Les enzymes (et les nutriments) sont transportés à la vitesse  $c$ , qui peut dépendre de  $x$  et  $t$  :

$$\frac{\partial z}{\partial t} = -c \frac{\partial z}{\partial x} - \bar{k}z. \quad (1)$$

Le premier terme exprime le transport, le deuxième que la quantité d'enzymes décroît exponentiellement en fonction du temps.

Les nutriments sont transportés et dégradés par les enzymes :

$$\frac{\partial A}{\partial t} = -c \frac{\partial A}{\partial x} - k(ph(x))Az \quad (2)$$

Le deuxième terme exprime que la proportion de  $A$  dégradée est proportionnelle à  $A$  (*loi d'action de masse*) à la quantité d'enzymes  $z$  avec une efficacité qui dépend du  $pH$  de l'intestin à l'endroit considéré.

Il y a transport par le flot, production et absorption (par la paroi intestinale) des aliments digérés :

$$\frac{\partial B}{\partial t} = -c \frac{\partial B}{\partial x} + k(ph(x))Az - \tilde{k}B \quad (3)$$

Comme précédemment le premier terme représente le transport, le deuxième représente les digérés créés par  $A$  et le dernier terme l'absorption des digérés par la paroi intestinale au taux  $\tilde{k}$ .

Vu le caractère hyperbolique de l'équation, on n'a besoin que des conditions aux limites en  $x = 0$ .

- ♦  $A(0, t) = \phi(t)$  sera une fonction périodique, typiquement une indicatrice.
- ♦  $B(0, t) = 0$  : pas de digestion instantanée, tout est transporté.
- ♦  $z(0, t) = \bar{z}\phi(t)$  : on suppose que les enzymes sont injectés en même temps que  $A$ .

Nous avons utilisé **Maple** pour résoudre les équations, mais ce modèle n'est pas facile à résoudre numériquement : les pas de temps et d'espace à considérer (de l'ordre de la dizaine de secondes et du centimètre) sont négligeables devant les dimensions du modèle : par exemple les 18 mètres de l'intestin grêle du porc sont parcourus en plusieurs heures.

## 3 Equations différentielles ordinaires

Face aux inconvénients du modèle précédent, nous avons mis au point un modèle de type différent basé sur des équations différentielles couplées : les unes s'occupant du transport et les autres de la dégradation.




FIG. 1 – Activité enzymatique

### 3.1 Premiers essais

Nos premiers essais ont encore été réalisés à l'aide de **Maple**<sup>TM</sup>.

Le calcul de l'activité enzymatique (fonction  $k(ph(x), z)$ ) est maintenant basé sur les hypothèses suivantes

- ◆ L'activité enzymatique est maximum (100%) pour un  $pH$  égal à 5 et nulle pour un  $pH$  égal à 2 ou à 7 avec variation linéaire entre ces points.
- ◆ Le  $pH$  de l'intestin varie (linéairement) en fonction de la distance entre 2 et 6 sur le premier mètre de l'intestin puis entre 6 et 7 sur les 17 mètres suivants.
- ◆ Le composé des deux (au sens de la composition des fonctions) donne une fonction « activité » en fonction de la distance qui passe par les points  $x = 0, k = 0$ ,  $x = 3/4, k = 1$ ,  $x = 1, k = 1/2$  et  $x = 18, k = 0$ .

D'où l'expression

$$k(x) = \frac{2}{3}|x| - \frac{5}{3}|x - \frac{3}{4}| + \frac{67}{68}|x - 1| + \frac{1}{68}|x - 18|.$$

que l'on retrouvera dans tous les programmes ci-dessous. La représentation graphique est donnée figure 1.

Le « paquet » de nutriments étant implicitement assimilé à une sphère de rayon  $R(t)$ , le premier modèle comportait 3 inconnues,  $x(t)$  la position,  $R(t)$  et  $z(t)$  la quantité d'enzymes actifs (mélangés aux nutriments).

$$\begin{cases} \frac{dx}{dt} &= c_0 + c_1 R(t) \\ \frac{dR}{dt} &= -2k(x(t))z(t)R(t) \\ \frac{dz}{dt} &= -\bar{k} z(t) \end{cases} \quad (4)$$

La première relation indique que la vitesse est une fonction affine du rayon, la troisième la décroissance exponentielle de l'activité enzymatique en fonction du temps. La deuxième exprime une *loi d'action de masse* : la diminution du rayon est fonction de l'activité enzymatique proportionnelle à la quantité d'enzymes actifs multipliée par la fonction d'activité.

Pour des raisons de portabilité et de facilité d'usage les modèles ultérieurs ont tous été implémentés à l'aide de **Scilab**<sup>TM</sup>. Une première amélioration a été de rajouter une variable correspondant aux « digérés » (ou digestibles, la distinction n'est pas claire) produits à partir des nutriments.

Le listing 1 est la réécriture des équations (4). La ligne 9 correspond au calcul de la fonction d'activité en fonction du  $pH$  exposé ci-dessus.

Listing 1 – dig.sci

---

```
1 function f=dig(t,x)
```

```

2   f(1)=c0+c1*x(2); // derivee de la position
3   f(2)=-ph(x(1))*x(3)*x(2); // derivee des nutriments
4   f(3)=-k*x(3); // derivee des enzymes
5   f(4)=ph(x(1))*x(3)*x(2)-kt*x(4); // derivee de B
6   endfunction
7
8   function kph=ph(x)
9       kph=2*abs(x)/3-5*abs(x-3/4)/3+67*abs(x-1)/68+abs(x-18)/68;
10  endfunction

```

---

Le programme associé est donné au listing 2 avec les valeurs utilisées pour les constantes.

Listing 2 – digODE2.sce

---

```

1 // constantes du probleme
2 c0=2;
3 c1=5;
4 k=log(4/3);
5 kt=1/3;
6 // resolution ODE
7 init=[0;1;1;0];
8 t=0:0.01:5;
9 y=ode(init,0,t,dig);
10 // trace dans la fenetre 1
11 scf(1);
12 xbase(1);
13 plot2d(t,y(2,:),style=1);
14 plot2d(t,y(4,:),style=2);
15 legends(['Nutr','Digeres'],[1,2], 'ur')
16 xtitle('Digestion')

```

---

## 3.2 Modèle de base

On considère que les nutriments sont constitués de « paquets » qui vont avoir des évolutions indépendantes; on peut les traiter séparément. Dans la suite on ne considérera généralement qu'un seul « paquet ».

Un « paquet » donné est décrit, à l'instant  $t$ , par un centre  $x(t)$  et un rayon  $R(t)$  et assimilé à une sphère. Si on note  $A(t)$  la masse de matière contenue dans cette boule, on a donc

$$A(t) = \frac{4}{3}\pi R^3(t)\rho,$$

où  $\rho$  est la masse volumique de la matière.

La matière est transportée et digérée : le transport se traduit par une équation sur  $x$  du type :

$$\frac{dx}{dt}(t) = c_0 + c_1 A(t) \quad (5)$$

qui reflète l'idée que plus on a de matière plus elle avance vite. La digestion se fait via la surface de contact et on suppose que la vitesse de disparition de  $A$  est proportionnelle à la surface de la sphère et

à l'activité enzymatique :

$$\begin{aligned}\frac{dA}{dt}(t) &= -k(ph(x(t)), z(t)).4\pi R^2(t) \\ &= -\left(\frac{36\pi}{\rho^2}\right)^{1/3} k(ph(x(t)), z(t)).A^{2/3}(t) .\end{aligned}\quad (6)$$

où  $z$  est la concentration en enzyme au point  $x(t)$  qui suit toujours la loi  $\frac{dz}{dt}(t) = -\bar{k}z(t)$ . Bien évidemment, il y a création de nutriments assimilables  $B(t)$  qui sont absorbés par la paroi avec un taux  $\tilde{k}$

$$\frac{dB}{dt}(t) = \left(\frac{36\pi}{\rho^2}\right)^{1/3} k(ph(x(t)), z(t)).A^{2/3}(t) - \tilde{k}B(t) . \quad (7)$$

On fait l'hypothèse que la matière avance à cause d'impulsions qui sont générées à l'entrée de l'intestin grêle et qui vont se propager dans tout l'intestin à vitesse  $c$ ; la durée de migration étant de 150 minutes pour les 18 m, on a donc une valeur de  $c$  de l'ordre de 7,2 m/h.

Les impulsions sont modélisées via une fonction  $y(t)$ ; les impulsions elles-mêmes sont données par la dérivée de  $y$ . On construit  $y$  de la manière suivante : on note  $\psi$  la fonction définie par :

$$\psi(t) = \begin{cases} 0 & \text{si } t \leq 0 \\ t & \text{si } 0 \leq t \leq 1 \\ 1 & \text{si } t \geq 1 \end{cases}$$

L'équation sur  $x$  sera de la forme (sans tenir compte de l'effet de  $A$  pour l'instant) :

$$\frac{d^2x}{dt^2}(t) = \frac{d}{dt} \left[ y\left(t - \frac{x(t)}{c}\right) \right] - K \frac{dx}{dt}(t) .$$

Le premier terme tenant compte des impulsions qui se propagent : en effet  $\frac{x(t)}{c}$  est le temps qu'a mis une impulsion pour arriver en  $x(t)$  et donc le signal que l'on doit prendre en compte est celui qui est généré à l'entrée de l'intestin au temps  $(t - \frac{x(t)}{c})$ ; le terme en  $K$  est un terme d'amortissement. Pour tenir compte du  $A$ , on écrit comme précédemment que plus le volume est grand plus l'impulsion est efficace :

$$\frac{d^2x}{dt^2}(t) = \frac{d}{dt} \left[ y\left(t - \frac{x(t)}{c}\right) \right] \frac{c_0 + c_1 A(t)}{a + bx(t)} - K \frac{dx}{dt}(t) ,$$

avec :

$$x(0) = 0 \quad , \quad \frac{dx}{dt}(0) = 0 .$$

Le listing 3 montre l'implémentation en **Scilab**<sup>TM</sup>.

#### Listing 3 – dig2.sci

---

```

1 // Modele avec vitesse de type impulsion
2 // constante K attenuation de la vitesse
3 // constante c c0 c1 a b voir modele vitesse
4 // constante kb disparition des enzymes
5 // constante kt absorption des nutriments.
6 //
7 // x(1) = position
8 // x(2) = derivee position (ie vitesse)
9 // x(3) = nutriments
10 // x(4) = enzymes
```




FIG. 2 – Fonction  $W(t)$ ,  $\alpha = 1/4$   $\tau = 6$   $t \in [0; 1,5]$

```

11 // x(5) = digeres
12 function f=dig2(t,x)
13     f(1)=x(2); // vitesse
14     f(2)=-K*x(2) + W(t-x(1)/c)*(c0+c1*x(3))/(a+b*x(1))*(1-x(2)/c) ; // acceleration
15     f(3)=-C*ph(x(1))*x(4)*abs(x(3))^(2/3); // derivee nutriments
16     f(4)=-kb*x(4); // derivee enzymes
17     f(5)=C*ph(x(1))*abs(x(3))^(2/3)*x(4)-kt*x(5); // derivee digeres
18 endfunction
19
20 // echelon de longueur alpha entre 0 et 1
21 function y=ech(x)
22     y=(alpha+abs(x)-abs(x-alpha))/2/alpha;
23 endfunction
24
25 // derivee de la precedente
26
27 function z=indic(x)
28     if x < 0 then z=0;
29     elseif x < alpha then z=1/alpha; else z= 0; end;
30 endfunction
31
32 // echelon periodique
33 function yw=W(x)
34     yw = tau*indic(tau*x-fix(tau*x));
35 endfunction

```

Les trois dernières fonctions servent à définir un échelon d'aire unité  $W$  qui se reproduit  $\tau$  fois par unité de temps et de largeur  $\alpha\tau$  c.f. figure 2.  $W$  est la dérivée de  $\psi$  complétée par périodicité. Pour la fonction de digestion, la ligne 14 résulte en fait de la dérivation d'une équation sur la vitesse. En dehors des périodes d'impulsion, la vitesse décroît exponentiellement (paramètre  $K$ ); pendant les périodes d'impulsion, le « paquet » est accéléré proportionnellement à une fonction affine du volume des nutriments (facteur  $c0 + c1 * x(3)$ ) avec une efficacité inversement proportionnelle à la distance déjà parcourue (facteur  $1/(a + b * x(1))$ ). Le terme  $(1 - x(2)/c)$  provient de la dérivée de la fonction composée  $W(t - x(1)/c)$ .

La ligne suivante exprime que la dégradation des nutriments est surfacique, d'où la puissance  $2/3$  (pour éviter les problèmes numériques on prend la valeur absolue). La constante  $C$  correspond à  $36\pi/\rho^2 \dots$

Le programme associé est donné au listing 4.

#### Listing 4 – digODEvg.sce

```

1 // L'equation est definie dans le fichier dig2.sci
2 // constantes du probleme

```

```

3 // constante K atténuation de la vitesse
4 K = 1.5;
5 // constante c c0 c1 a b voir modele
6 a = 2.72;
7 b= 1.2346;
8 c= 7.0;
9 c0 = 1.0;
10 c1 = 2.0;
11 // constante kb disparition des enzymes
12 kb=log(4/3);
13 // constante kt absorption des nutriments.
14 kt=1/3;
15 // constante C correspond a  $36\pi/\rho^2$ ....
16 C = 1.5;
17 alpha = 0.1; //duree impulsion = 1mn
18 tau = 6.0; // espacement impulsions = 10 mn
19 // resolution ODE
20 init=[0;1;1;1;0];
21 t=0:0.005:5;
22 y=ode("rkf",init,0,t,dig2);
23 // trace dans la fenetre 0 de x et dx/dt
24 scf(0);
25 xbas(0);
26 xtitle('x et dx/dt vs t')
27 plot2d(t,y(1,:),style=1);
28 plot2d(t,y(2,:),style=2);
29 legends(['x','dx/dt'],[1,2],"ur")
30 // trace dans la fenetre 1
31 scf(1);
32 xbas(1);
33 plot2d(t,y(3,:),style=1);
34 plot2d(t,y(5,:),style=2);
35 legends(['Nutr','Digeres'],[1,2],"ur")
36 xtitle('Digestion')

```

---

Voici les résultats obtenus


Nous avons aussi implanté une version « multi-paquets », où on fait entrer plusieurs « paquets » de nutriments.

Si  $V(t)$  est le volume de l'estomac à l'instant  $t$  alors on suppose que les volumes injectés décroissent exponentiellement  $V(t) = \exp(-\alpha t)V(0)$  On choisit des temps d'ouverture  $T_0 = 0 < T_1 < T_2 \dots < T_n$  (a priori  $T_{i+1} - T_i$  est constant) et à l'instant  $T_i$ , l'estomac délivre une quantité de matière égale à  $V(T_i) - V(T_{i+1})$ , convertie avec le bon rayon ou la bonne masse suivant :


$$A(x=0, T_i) = \rho(V(T_i) - V(T_{i+1})) \quad \rho \text{ étant la masse volumique de la matière.}$$

Ceci génère une sphère de centre  $x_i(t)$  et de rayon  $R_i(t)$  se propageant à l'intérieur de l'intestin grêle. Comme les sphères sont de tailles décroissantes elles se propagent donc à des vitesses qui vont aussi en décroissant et ne peuvent donc pas se rencontrer : elles se propagent indépendamment. La vitesse d'injection  $dx_i/dt(T_i)$  est proportionnelle à ce qui reste dans l'estomac :

$$\frac{dx_i}{dt}(T_i) = \beta V(T_i) .$$



(a) Vitesse et Position



(b) Nutriments et Digérés

FIG. 3 – Evolution par impulsions d'un seul paquet

Le programme correspondant est donné au listing 5. Aux lignes 14-24 on définit la matrice des conditions initiales, le coefficient  $\alpha$  et renommé *alfa* pour le distinguer du  $\alpha$  qui définit la largeur des impulsions. Les temps d'injections correspondent aux impulsions,  $\tau$  fois par heures ; ce qui revient à dire qu'une vidange gastrique déclenche une impulsion. A la ligne 57 on cherche à estimer le temps auquel le « paquet » est sorti et sa composition, que l'on fait afficher ensuite. On termine par un cumul sur l'ensemble des paquets.

Listing 5 – digvgloop.sce

---

```

1 K = 0.1;
2 a = 2.72;
3 b= 1.2346;
4 c=7;
5 c0 = 1.0;
6 c1 = 2.0;
7 kb=log(4/3);
8 kt=1/3;
9 C = 1.5;
10 alpha = 1.0/10; //duree impulsion = 1mn (1/10 de l'espaceement).
11 tau = 6.0; // inverse espaceement impulsions = 10 mn
12 xbasc(0);xbasc(1);
13 // ***** conditions initiales *****
14 Tmax = 1; // injection pendant une heure
15 alfa = input("Coefficient_alpha?"); // decroissance des injectes
16 Ti = [0:1/tau:Tmax]; // temps des injections
17 Pos = zeros(Ti);
18 Vit = exp(-alfa*Ti);
19 Nut = (Vit - [Vit(2:$) 0])*2.0; // calcul de V(Ti)-V(Ti+1)
20 Enz = ones(Ti);
21 Dig = zeros(Ti);
22 // matrice des conditions initiales pour l'ensemble des injectes
23 Minit = [Ti;Pos;Vit;Nut;Enz;Dig];
24 //
25 printf("*****_debut_*****\n");

```

```

26 // initialisation cumuls et couleur trace
27 nutcum=0;
28 digcum=0;
29 couleur = 1;
30 // Boucle (principale) sur les conditions initiales
31 for toto=Minit do
32 // resolution ODE
33     init = toto(2:6);
34     tzero = toto(1);
35 // etude sur 6 heures a partir de tzero pour etre certain que le paquet
36 // atteigne les 15 metres
37     t=tzero:0.01:6+tzero;
38 // methode rkf obligatoire pour eviter les artefacts : en effet
39 // si on utilise une methode avec adaptation du pas, certaines impulsions
40 // ne sont pas prises en compte.
41     y=ode("rkf",init,tzero,t,dig2);
42 // trace dans la fenetre 0 de x et dx/dt
43     scf(0);
44     xtitle('x et dx/dt vs t')
45     plot2d(t,y(1,:),style=couleur);
46     plot2d(t,y(2,:),style=couleur+1);
47     legends(['x','dx/dt'],[1,2],"ur")
48 // trace dans la fenetre 1
49     scf(1);
50     plot2d(t,y(3,:),style=couleur);
51     plot2d(t,y(5,:),style=couleur+1);
52     legends(['Nutr','Digeres'],[1,2],"ur")
53     xtitle('Digestion')
54 // instructions en fin d'iteration
55     couleur = couleur+2;
56     printf("Entree = %f\n",toto(4));
57 // affichage sortie (15 metres)
58     tsv = find(y(1,*)>15); // vecteur des temps
59 // ou on est dehors : seul compte le premier !
60     if length(tsv)>0 then //on est sorti
61         ts = tsv(1)*0.01+tzero;
62         nutcum = nutcum+y(3,tsv(1));
63         digcum = digcum +y(5,tsv(1));
64         printf("Sortie a t=%f\n tNutriments=%f\n tDigeres = %f\n\n",ts,y(3,tsv(1)),y(5,tsv(1)));
65     else printf("Pas sorti\n\n")
66     end //endif
67 //sortie boucle
68 end
69 // Resume
70 printf("\nCumules\n tNutriments=%f\n tDigeres=%f\n\n",nutcum,digcum)
71 // fin .


```

---


Un exemple d'exécution produit la sortie suivante; les graphiques correspondants sont donnés par la figure 4

Coefficient alpha? 4.5

\*\*\*\*\* debut \*\*\*\*\*



(a) Vitesse et Positions



(b) Nutriments et Digérés

FIG. 4 – Evolution de chacun des paquets

|                     |                     |                     |                     |
|---------------------|---------------------|---------------------|---------------------|
| Entree = 1.055267   | Entree = 0.498473   | Entree = 0.235462   | Entree = 0.111224   |
| Sortie a t=4.21     | Sortie a t=5.48667  | Sortie a t=6.323333 | Pas sorti           |
| Nutriments 0.192504 | Nutriments 0.024200 | Nutriments 0.000587 |                     |
| Digérés = 0.312159  | Digérés = 0.120240  | Digérés = 0.044843  |                     |
| Entree = 0.052539   | Entree = 0.024817   | Entree = 0.22218    | Cumules             |
| Pas sorti           | Pas sorti           | Pas sorti           | Nutriments=0.217291 |
|                     |                     |                     | Digérés=0.477242    |

### 3.3 Modèle Digérés-Digestibles

Nous avons apporté deux améliorations au modèle précédent. D'une part on distingue maintenant deux types de digérés : ceux qui sont directement assimilables et les autres. D'autre part on prend en compte les sécrétions biliaires et pancréatiques qui sont injectés vers la fin du *duodenum*.

#### 3.3.1 Digérés

On tient compte du fait que l'absorption a lieu en périphérie de l'intestin, et donc du « paquet » : dans un premier temps les nutriments  $A$  sont transformés en digérés  $B_1$  sous l'action des enzymes qui sont mélangés avec les nutriments, dans un deuxième temps le  $B_1$  situé en périphérie, noté  $B_2$ , est transformé en assimilables (i.e. disparaît dans la circulation sanguine) sous l'action des enzymes de la bordure en brosse (*brush-border*). A priori la différence entre  $B_1$  et  $B_2$  est uniquement spatiale (et non chimique) ; on suppose en outre que la concentration des enzymes *brush-border* est toujours suffisante pour hydrolyser tout le  $B_2$  produit.

Ceci conduit à remplacer le terme :

$$\left(\frac{36\pi}{\rho^2}\right)^{1/3} k(ph(x(t)), z(t)).A^{2/3}(t)$$

dans l'équation de  $A$  par :

$$C_1 k(ph(x(t)), z(t)).A(t) + C_2(x(t)).A^{2/3}(t) ,$$

où  $C_1$  est une constante.

Le premier terme prend en compte l'activité enzymatique à l'intérieur de la sphère que l'on suppose homogène : on conserve une dépendance par rapport à la concentration d'enzyme  $z(t)$  et, bien sûr, par rapport au  $pH$ . La constante  $C_1$  (ou une autre constante dans l'équation de  $B$ ) peut prendre en compte la part directement assimilable dans l'organisme.

On considère donc que  $B = B_1 + B_2$  où  $B_1$  représente la part de  $B$  qui reste à l'intérieur de la sphère et qui, donc, n'est pas assimilable par l'organisme et  $B_2$  est la part assimilable qui est en surface. On suppose que  $A, B_1, B_2$  ont la même masse volumique et qu'à l'intérieur de la sphère  $A$  et  $B_1$  sont répartis de manière homogène (même proportion en tout point). On note  $r = B_1/(A+B_1)$  la proportion de  $B_1$  dans la sphère et on a la relation  $A + B_1 = \rho V$  où  $V = 4\pi R^3$  est le volume de la sphère. La transformation de  $A$  en  $B_1, B_2$  se fait de deux façons :

- ♦ de manière « volumique »,  $A$  ne produit que du  $B_1$  suivant la loi

$$dA = -cte.(1 - r(t))R^3 = -cte.A(t) ,$$

et :

$$dB_1 = +cteA(t)$$

auquel il faudra enlever la partie transformée en  $B_2$  par l'effet de surface.

- ♦ de manière « surfacique » : le mélange  $A + B_1$  se transforme en  $B_2$  car  $B_1$  en surface est par définition du  $B_2$  et que l'activité des enzymes "brush borders" dégrade  $A$  en  $B_1 \equiv B_2$ .

On a un  $dA$  proportionnel à la surface de  $A$  à la périphérie de la sphère de rayon  $R$ , d'où si on note  $r = B_1/(A + B_1)$  la proportion de  $B_1$ , on a (seulement pour cet effet) :

$$dA = -cte.(1 - r(t))R^2 ,$$

car  $(1 - r(t))R^2$  représente la surface "occupée par  $A$ " à la périphérie de la sphère. Finalement, puisque  $CR^3 = A + B_1$ , on trouve :

$$dA = -cte.(1 - r(t))(A + B_1)^{2/3} = -cte.\frac{A}{(A + B_1)^{1/3}} .$$

Mais cette transformation de  $A$  (qui devient vraiment du  $B_2$ ) est associée à une libération en proportion de  $B_1$  qui devient du  $B_2$ . Et on a donc :

$$dB_2 = +cte.r(t)R^2 = +cte\frac{B_1}{(A + B_1)^{1/3}} .$$

Après cette présentation un peu formelle, les équations sur  $A, B_1, B_2$  deviennent :

$$\frac{dA}{dt}(t) = -C_1k(ph(x(t)), z(t)).A(t) - C_2(x(t)).\frac{A}{(A + B_1)^{1/3}(t)} \quad (8)$$

$$\frac{dB_1}{dt}(t) = +C_1k(ph(x(t)), z(t)).A(t) - C_2(x(t)).\frac{B_1}{(A + B_1)^{1/3}(t)}$$

$$\frac{dB_2}{dt}(t) = C_2(x(t)).(A + B_1)^{2/3}(t) - \tilde{k}B_2(t) ,$$

**NB :** Le premier terme est de la forme  $cte.R^2$  (somme de deux termes  $+cte.(1 - r(t))R^2$  et  $+cte.r(t)R^2$ ) donc l'équation sur  $R$  n'est pas vraiment modifiée : certes la surface occupée par  $A$  sur le bord de la sphère est moindre et donc la quantité transformée par les enzymes est moindre aussi mais on a le renfort du  $B_1$  qui est libéré et qui compense exactement cette moindre activité périphérique.

### 3.3.2 Secrétions

Dans ce même modèle on prend en compte les sécrétions pancréatiques : après une distance parcourue par le « paquet » de l'ordre de 5% de la longueur totale de l'intestin grêle (soit ici approximativement 0,85m), le bol alimentaire reçoit des sécrétions qui augmentent sa masse de 25%. A priori ces sécrétions sont essentiellement enzymatiques, mais dans certains cas elle apportent aussi des nutriments. Donc, si  $x(\bar{t}) = 0.85$  alors  $A(\bar{t}_+) = 1.25A(\bar{t}_-)$  [le “+” indique la valeur juste après l’instant  $\bar{t}$  et le “-” sa valeur juste avant. Pour éviter d’introduire une discontinuité on va faire rentrer les sécrétions sur quelques centimètres au lieu de le faire instantanément. Ceci revient à rajouter un terme dans l’équation (8 du type :

$$\dots + \log(1.25)l^{-1} \frac{dx}{dt}(t)A(t) \frac{d}{dt} \left[ \psi \left( \frac{x(t) - 0.85}{l} \right) \right] .$$

En effet, si ce terme agit seul (et il est raisonnable de le supposer car son effet est quasi-instantané) alors on est amené à résoudre :

$$\frac{dA}{dt}(t) = \log(1.25)l^{-1} \frac{dx}{dt}(t)A(t)$$

sur l’intervalle de temps  $(t_0, t_1)$  où  $t_0$  est le temps pour lequel  $x(t_0) = 0.85$  alors que  $t_1$  est celui pour lequel  $x(t_1) = 0.85+l$  (puisque tout se passe quand la position de la sphère est entre 0.85m et  $(0.85+l)$ m et qu’alors le terme  $\frac{d}{dt} \left[ \psi \left( \frac{x(t)-0.85}{l} \right) \right]$  vaut 1). L’intégration de cette équation donne :

$$A(t) = A(t_0) \exp \left[ \log(1.25) \frac{x(t) - x(t_0)}{l} \right] ,$$

et donc  $A(t_1) = 1.25A(t_0)$ .

Le listing 6 montre la nouvelle fonction de digestion.

Listing 6 – digAB1B2.sci

---

```

1 // Modele avec vitesse de type impulsion et secretions
2 // et deux types de digeres B1 et B2
3 // constante K attenuation de la vitesse
4 // constante c c0 c1 a b voir modele vitesse
5 // constante kb disparition des enzymes
6 // constante kt absorption des nutriments.
7 // constantes C1 et C2 pour degradation A vers B1 et B2
8 // x(1) = position
9 // x(2) = derivee position (ie vitesse)
10 // x(3) = nutriments
11 // x(4) = enzymes
12 // x(5) = digeres pas assimilables B1
13 // x(6) = digeres assimilables B2
14 function f=digABB(t,x)
15     f(1)=x(2);
16     f(2)=-K*x(2) + W(t-x(1)/c)*(c0+c1*x(3))/(a+b*x(1))*(1-x(2)/c) ;
17     // pour eviter les problemes on prend la valeur absolue
18     // avec ajout du terme de secretions
19     f(3)=-C1*ph(x(1))*x(4)*x(3) - C2*ph(x(1))*x(3)/abs(x(3)+x(5))^(1/3)..
20         +1.25*x(3)*indic(x(1)-0.85);
21     f(4)=-kb*x(4);
22     f(5)=C1*ph(x(1))*x(3)*x(4)- C2*ph(x(1))*x(5)/abs(x(3)+x(5))^(1/3);

```

```

23     f(6)= C2*ph(x(1))*abs(x(3)+x(5))^(2/3)-kt*x(6);
24 endfunction

```

---

Le listing 7 montre l'exécutable associé. Nous avons amélioré l'affichage des résultats en faisant figurer sur chaque graphique les valeurs de la plupart des constantes utilisées.


Listing 7 – digABBx.sce

---


```

1  // Variables du probleme :
2  // L'equation est definie dans le fichier digAB1B2.sci
3  // constantes du probleme
4  c0=15;
5  c1=7.5;
6  k=log(4/3);
7  kt=0.0; // on neglige l'absorption ?
8  a=2.72;
9  b=1.2346;
10 c=8.0;
11 kb=log(4/3);
12 C1=1.5;
13 C2=2.5;
14 alpha=1.0/3;
15 tau=15.0;
16 K = 3.0;
17 init=[0;1;1;1;0.5;0];
18 t=0:0.01:5;
19 y=ode("rkf",init,0,t,digABB);
20 // ***** affichage des constantes *****
21 ctes_nom = ["K_=", msprintf('%4.2 f ',K)
22             "a_=", msprintf('%4.2 f ',a)
23             "b_=", msprintf('%4.2 f ',b)
24             "c_=", msprintf('%4.2 f ',c)
25             "c0=", msprintf('%4.2 f ',c0)
26             "c1=", msprintf('%4.2 f ',c1)
27             "kb=", msprintf('%4.2 f ',kb)
28             "C1=", msprintf('%4.2 f ',C1)
29             "C2=", msprintf('%4.2 f ',C2)
30             "kt=", msprintf('%4.2 f ',kt)];
31 // *****
32 // trace dans la fenetre 0
33 scf(0);
34 xbas(0);
35 plot2d(y(1,:),y(3,:),style=1); // A
36 plot2d(y(1,:),y(5,:),style=2); // B1
37 plot2d(y(1,:),y(6,:),style=3); // B2
38 plot2d(y(1,:),y(3,:)+y(5,:),style=4);
39 //plot2d(t,y(3,:)+y(5,:)+y(6,:),style=5);
40 legends(['Nutr','B1','B2','A+B1'],[1,2,3,4],"ur");
41 xtitle('Digestion_vs_distance_en_m');
42 [w1,w2,w3,w4]=xgetech();
43 xstring(9*w2(3)/10,w2(4)/4,ctes_nom);
44 scf(1);

```




(a) Vitesse et Position




(b) Nutriments et Digérés

FIG. 5 – Courtes impulsions rapprochées



(a) Vitesse et Position



(b) Nutriments et Digérés

FIG. 6 – Impulsions plus espacées

```

45 xbasc(1);
46 plot2d(t,y(2,:),style=2);
47 plot2d(t,y(1,:),style=3);
48 xtitle(' Vitesse et position vs temps (en m/h) ');
49 legends([' Vitesse ', ' Position '],[2,3], "ul");
50 [w1,w2,w3,w4]=xgetech();
51 xstring(9*w2(3)/10,w2(4)/4,ctes_nom);

```

Un premier jeu de résultats correspondant à de courtes impulsions ( $\alpha = 1/3$ ) assez rapprochées ( $\tau = 15$ ) est représenté figure 5.

Un deuxième jeu de résultats correspondant à des impulsions plus longues et plus espacées ( $\alpha = 0,8, \tau = 8$ ) est représenté figure 6.

Dans les deux cas, on peut observer l'accroissement des nutriments provoqué par les sécrétions : on peut aussi remarquer que l'accroissement n'est pas exactement égal à 25%. Nous reviendrons sur ce point au paragraphe suivant.

### 3.4 Modèle à cylindres

Nous avons recruté à l'été 2008 une stagiaire N. Traoré qui à repris, dans un premier temps, le code **Scilab** pour le vérifier, puis a mis au point un changement de géométrie des « paquets » qui sont maintenant assimilés à des cylindres de longueur fixe  $\ell$  mais de rayon variable  $r$ .

Soit  $V$  le volume du cylindre,  $\rho$  la masse volumique de la matière. Le volume est constitué de  $A + B_1$  car  $B_2$  étant réparti en surface et absorbé instantanément ne compte pas dans le volume. On a

$$V = \frac{1}{\rho}(A + B_1) = \pi r^2 \ell \quad (9)$$

D'où  $r = \sqrt{A + B_1} / \sqrt{\pi \ell \rho}$ . La réaction  $A \rightarrow B_1$  dépend de l'activité  $Z(x)$  et de la concentration en enzymes  $z/z_0$ ; la vitesse de réaction est donnée par :

$$\frac{d}{dt}\left(\frac{\rho A}{A + B_1}\right) = -C_1 Z(x) \frac{z}{z_0} \frac{A}{A + B_1} \quad (10)$$

en considérant les concentrations volumiques des réactifs et produits.  $C_1$  est une constante de vitesse et  $z_0$  la concentration initiale en enzymes. A partir de

$$\frac{d}{dt}\left(\frac{\rho A}{A + B_1}\right) = \frac{\rho}{A + B_1} \frac{dA}{dt} - \frac{\rho A}{(A + B_1)^2} \frac{d(A + B_1)}{dt},$$

on obtient

$$\frac{dA}{dt} = -\frac{C_1}{\rho} Z(x) \frac{z}{z_0} A + \frac{A}{A + B_1} \frac{d(A + B_1)}{dt}. \quad (11)$$

Le premier terme représente la variation de  $A$  due à la réaction  $A \rightarrow B_1$  et donc l'opposé de la variation de  $B_1$ .

Les variations de masse dues aux réactions  $A \rightarrow B_2$  et  $B_1 \rightarrow B_2$  s'écrivent

$$\frac{dA}{dt} = -C_2 2\pi r \ell \frac{A}{A + B_1} = -2C_2 \sqrt{\frac{\pi \ell}{\rho}} \frac{A}{\sqrt{A + B_1}}, \quad (12)$$

$$\frac{dB_1}{dt} = -C_2 2\pi r \ell \frac{B_1}{A + B_1} = -2C_2 \sqrt{\frac{\pi \ell}{\rho}} \frac{B_1}{\sqrt{A + B_1}}, \quad (13)$$

où  $C_2$  est la vitesse de création de  $B_2$ . On obtient ainsi les parts respectives de  $B_2$  produites par  $A$  et  $B_1$ . On en déduit la relation

$$\frac{d(A + B_1)}{dt} = -2C_2 \sqrt{\frac{\pi \ell}{\rho}} \sqrt{A + B_1}. \quad (14)$$

En reportant dans (11) on obtient d'abord


$$\frac{dA}{dt} = -\frac{C_1}{\rho} Z(x) \frac{z}{z_0} A - 2C_2 \sqrt{\frac{\pi \ell}{\rho}} \frac{A}{\sqrt{A + B_1}},$$

puis à l'aide de (14)


$$\frac{dB_1}{dt} = \frac{C_1}{\rho} Z(x) \frac{z}{z_0} A - 2C_2 \sqrt{\frac{\pi \ell}{\rho}} \frac{B_1}{\sqrt{A + B_1}}.$$

Comme la variation de volume est due uniquement à la production de  $B_2$ , on obtient

$$\frac{dV}{dt} = 2\pi r \ell \frac{dr}{dt} = \frac{1}{\rho} \frac{d(A + B_1)}{dt},$$



(a) Vitesse et Positions



(b) Nutriments et Digérés

FIG. 7 – Modèle cylindre à longueur constante.

d'où on tire

$$\frac{dr}{dt} = -\frac{C_2}{\rho}. \quad (15)$$

Enfin la relation sur  $B_2$  est obtenue à partir de

$$\frac{dB_2}{dt} = -\rho \frac{dV}{dt} - \tilde{k}B_2 = 2C_2 \sqrt{\frac{\pi \ell}{\rho}} \frac{B_1}{\sqrt{A+B_1}} - \tilde{k}B_2. \quad (16)$$

Ces équations sont mises en œuvre au listing 8 et un exemple d'exécution à la figure 7.

Listing 8 – digcylindre2.sci

```

1 // Modele avec vitesse de type impulsion
2 // deux types de digeres B1 et B2 et
3 //
4 // cylindre de rayon r et de longueur l (l cste)
5 //
6 // constante K attenuation de la vitesse
7 // constante c c0 c1 a b voir modele vitesse
8 // constante kb disparition des enzymes
9 // constante kt absorption des nutriments.
10 // constantes C1 pour degradation A vers B1, et C2 pour A et B1 vers B2
11 // x(1) = position
12 // x(2) = derivee position (ie vitesse)
13 // x(3) = nutriments (A)
14 // x(4) = enzymes
15 // x(5) = digeres pas assimilables B1
16 // x(6) = digeres assimilables B2
17 // x(7) = rayon du cylindre
18 function f=digABB(t,x)
19     f(1)= x(2);
20     f(2)= - K*x(2) + W(t-x(1)/c)*(c0+c1*x(3))/(a+b*x(1))*(1-x(2)/c) ;
21     // pour eviter les problemes on prend la valeur absolue
22     // avec ajout du terme de secretion

```

```

23 f(3)= - C2*2*sqrt(Pi*l/rho)*x(3)/sqrt(abs(x(3)+x(5)))..
24 - C1/rho*ph(x(1))*x(4)/z0*x(3)..
25 +log(1.25)/alpha*x(2)*(x(3)+x(5))*indic((x(1)-0.85)/alpha);
26 f(4)= - kb*x(4)+log(1.25)/alpha*x(2)*x(4)*indic((x(1)-0.85)/alpha); //saut enzyme
27 f(5)= - C2*2*sqrt(Pi*l/rho)*x(5)/sqrt(abs(x(3)+x(5)))..
28 + C1/rho*ph(x(1))*x(4)/z0*x(3);
29 f(6)= C2*2*sqrt(Pi*l/rho)*sqrt(abs(x(3)+x(5))) - kt*x(6);
30 f(7)= - C2/rho + log(1.25)/alpha/2*x(2)*indic((x(1)-0.85)*10)*x(7);
31 endfunction

```

---

Le calcul du terme de sécrétion a été repris de la façon suivante : soit  $t$  tel que  $x(t) = 0,85$  et  $t^+$  tel que  $x(t^+) = 0,85 + \alpha^2$ . Les sécrétions sont injectées dans l'intervalle  $[0,85; 0,86]$  si on choisit  $\alpha = 0,1$ . On doit avoir

$$V(t^+) = 1,25V(t), \quad r(t^+) = \sqrt{1,25}r(t), \quad (A + B_1)(t^+) = 1,25(A + B_1)(t).$$

En supposant  $B_1$  constante sur  $[t; t^+]$ , l'équation différentielle suivante

$$\frac{dr}{dt} = \frac{1}{2} \ln(1,25) \frac{v}{\alpha} \text{indic}\left(\frac{x - 0,85}{\alpha}\right)r$$

répond aux conditions, avec la fonction `indic()` définie à la ligne 27 du listing 3. En effet, par intégration, on obtient

$$r(t^+) = \exp\left(\frac{1}{2} \ln(1,25) \int_t^{t^+} \frac{v(s)}{\alpha} \text{indic}\left(\frac{x(s) - 0,85}{\alpha}\right) ds\right) r(t), \quad (17)$$

$$= \exp\left(\frac{1}{2} \ln(1,25) \left(\phi\left(\frac{x(t^+) - 0,85}{\alpha}\right) - \phi\left(\frac{x(t) - 0,85}{\alpha}\right)\right)\right) r(t), \quad (18)$$

où  $\phi$  désigne la primitive de `indic()` nulle en 0. On a donc  $r(t^+) = \sqrt{1,25}r(t)$  comme souhaité. De la même façon on trouve

$$\frac{dA}{dt} = \ln(1,25) \frac{v}{\alpha} \text{indic}\left(\frac{x(s) - 0,85}{\alpha}\right)(A + B_1), \quad (19)$$

$$\frac{dz}{dt} = \ln(1,25) \frac{v}{\alpha} \text{indic}\left(\frac{x(s) - 0,85}{\alpha}\right)z \quad (20)$$

Nous avons apporté deux améliorations à ce programme :

- ◆ Calcul et affichage des quantités cumulées pour  $A$ ,  $B_1$  et  $B_2$  par intégration.
- ◆ Introduction d'une boîte de dialogue pour éditer les valeurs des constantes du problème.

Les modifications apportées à la fonction `digABB` apparaissent dans le listing 9 : les trois dernières lignes correspondent à l'intégration des quantités respectives  $A$ ,  $B_1$  et  $B_2$ .

Listing 9 – digcylindre3sci

```

1 function f=digABB(t,x)
2   f(1)= x(2);
3   f(2)= - K*x(2) + W(t-x(1)/c)*(c0+c1*x(3))/(a+b*x(1))*(1-x(2)/c) ;
4   // pour eviter les problemes on prend la valeur absolue
5   // avec ajout du terme de boost
6   f(3)= - C2*2*sqrt(Pi*l/rho)*x(3)/sqrt(abs(x(3)+x(5))) - C1/rho*ph(x(1))*x(4)/z0*x(3)

```




FIG. 8 – Choix des constantes et affichage

```

7      +log(1.25)/alpha*x(2)*(x(3)+x(5))*indic((x(1)-0.85)/alpha);
8      f(4)= - kb*x(4)+log(1.25)/alpha*x(2)*x(4)*indic((x(1)-0.85)/alpha); //saut enzyme
9      f(5)= - C2*2*sqrt(Pi*l/rho)*x(5)/sqrt(abs(x(3)+x(5)))..
10         + C1/rho*ph(x(1))*x(4)/z0*x(3);
11      f(6)= C2*2*sqrt(Pi*l/rho)*sqrt(abs(x(3)+x(5))) - kt*x(6);
12      f(7)= - C2/rho + log(1.25)/alpha/2*x(2)*indic((x(1)-0.85)/alpha)*x(7);
13      f(8)= C1/rho*ph(x(1))*x(4)/z0*x(3);
14      f(9)= C2*2*sqrt(Pi*l/rho)*x(3)/sqrt(abs(x(3)+x(5)));
15      f(10)=C2*2*sqrt(Pi*l/rho)*x(5)/sqrt(abs(x(3)+x(5)));

```

Le programme correspondant est donné au listing 10 et un exemple d'exécution à la figure 8.

Listing 10 – digcylindrex3sce

```

1 // Variables du probleme :
2 // L'equation est definie dans le fichier digcylindre3.sci
3 // constantes du probleme
4
5 entry=['K','c0','c1','a','b','c','C1','C2',...
6       'kt','kb','l','rho','alpha','tau',...
7       'x0','v0','z0','B10','B20','r0'];

```

```

8  init=[ '15 ', '15 ', '1000 ', '2.72 ', '1.2346 ', '8 ', '1200 ', '1 ', ...
9         '.5 ', 'log(4/3) ', '.1 ', '1000 ', '.1 ', '5 ', ...
10        '0 ', '1 ', '.001 ', '0 ', '0 ', '.032 '];
11
12  Pi=4*atan(1);
13
14
15  t=0:0.005:7;
16  while %t do
17  donnee= x_mdialog( 'Parametres',entry,init );
18  if donnee == [] then break; end
19  init=donnee;
20
21  K=evstr(donnee(1));
22  c0=evstr(donnee(2));
23  c1=evstr(donnee(3));
24  a=evstr(donnee(4));
25  b=evstr(donnee(5));
26  c=evstr(donnee(6));
27  C1=evstr(donnee(7));
28  C2=evstr(donnee(8));
29  kt=evstr(donnee(9));
30  kb=evstr(donnee(10));
31  l=evstr(donnee(11));
32  rho=evstr(donnee(12));
33  alpha=evstr(donnee(13));
34  tau=evstr(donnee(14));
35
36  //conditions initiales
37  x0=evstr(donnee(15));
38  v0=evstr(donnee(16));
39  z0=evstr(donnee(17));
40  B10=evstr(donnee(18));
41  B20=evstr(donnee(19));
42  r0=evstr(donnee(20));
43
44  A0=rho*Pi*l*(r0)^2;
45  B1A0=B_10;
46  B2A0=B_20;
47  B2B10=B_20;
48
49  CI=[x0;v0;A0;z0;B10;B20;r0;B1A0;B2A0;B2B10];
50  y=ode("rkf",CI,0,t,digABB);
51  // ***** affichage des constantes *****
52  ctes_nom = ["K_=", msprintf( '%4.2 f ',K)
53             "a_=", msprintf( '%4.2 f ',a)
54             "b_=", msprintf( '%4.2 f ',b)
55             "c_=", msprintf( '%4.2 f ',c)
56             "c0=", msprintf( '%4.2 f ',c0)
57             "c1=", msprintf( '%4.2 f ',c1)
58             "kb=", msprintf( '%4.2 f ',kb)

```

```

59         "C1=" , msprintf( '%4.2 f ' ,C1)
60         "C2=" , msprintf( '%4.2 f ' ,C2)
61         "kt=" , msprintf( '%4.2 f ' ,kt ) ] ;
62 // *****
63 // trace dans la fenetre 0
64 scf(0);
65 xbas(0);
66 plot2d(y(1,:),y(3,:),style=1); // A
67 plot2d(y(1,:),y(5,:),style=2); // B1
68 plot2d(y(1,:),y(6,:),style=3); // B2
69 plot2d(y(1,:),y(3,:)+y(5,:),style=5); //A+B1
70 //plot2d(y(1,:),rho*Pi*l*(y(7,:))^2,style=4);
71 plot2d(y(1,:),y(7,:),style=4); //r
72 plot2d(y(1,:),y(3,:)+y(5,:)+y(6,:),style=6); //A+B1+B2
73 legends([ 'A' , 'B1' , 'B2' , 'A+B1' , 'A+B1+B2' , 'r' ] , [1,2,3,5,6,4] , "ur" );
74 xtitle( ' Digestion \_vs\_ distance \_en \_m ' );
75 // [w1,w2,w3,w4]=xgetech();
76 // xstring(9*w2(3)/10,w2(4)/4,ctes_nom);
77
78 scf(1);
79 xbas(1);
80 plot2d(t,y(2,:),style=2);
81 plot2d(t,y(1,:),style=3);
82 xtitle( ' Vitesse \_et\_ position \_vs\_ temps \_(en \_m/h) ' );
83 legends([ 'Vitesse' , 'Position' ] , [2,3] , "ul" );
84 [w1,w2,w3,w4]=xgetech();
85 xstring(9*w2(3)/10,w2(4)/4,ctes_nom);
86
87 scf(2);
88 xbas(2);
89 plot2d(y(1,:),y(8,:),style=2); //quantite de B1
90 plot2d(y(1,:),y(9,:),style=5); //quantite de B2 direct
91 plot2d(y(1,:),y(10,:),style=6); //quantite de B2 qui vient de B1
92 legends([ 'B1_cree' , 'B2_par_A' , 'B2_par_B1' ] , [2,5,6]);
93 xtitle( 'Cummul_B' );
94
95 end

```

On peut se poser la question de savoir pourquoi, dans le cylindre, on choisit de fixer la longueur plutôt que le rayon ? On a donc détaillé ci-dessous, le même calcul en supposant cette fois-ci le rayon fixe et la longueur variable. Physiologiquement, nous ne savons pas quelle approche est la plus raisonnable.

Supposons donc que le cylindre est de rayon constant  $r$  et de longueur variable  $\ell$ . Les formules (9) donnant le volume et (11) donnant la dérivée de  $A$  restent valables. A partir de  $A + B_1 = \rho\pi r^2 \ell$  on obtient

$$\frac{d(A + B_1)}{dt} = \rho\pi r^2 \frac{d\ell}{dt}. \quad (21)$$

En reportant (9) dans les équations (12) et (13) on obtient les équations régissant la production de  $B_2$

$$\frac{dA}{dt} = -C_2 2\pi r \ell \frac{A}{A + B_1} = -2C_2 \frac{A}{\rho r}, \quad (22)$$

$$\frac{dB_1}{dt} = -C_2 2\pi r \ell \frac{B_1}{A + B_1} = -2C_2 \frac{B_1}{\rho r}, \quad (23)$$

d'où

$$\frac{d(A + B_1)}{dt} = -2C_2 \frac{A + B_1}{\rho r}.$$

La relation (11) s'écrit alors

$$\frac{dA}{dt} = -\frac{C_1}{\rho} Z(x) \frac{z}{z_0} A - A \frac{2C_2}{\rho r}. \quad (24)$$

De même

$$\frac{dB_1}{dt} = -\frac{dA}{dt} - 2C_2 \frac{A + B_1}{\rho r} = \frac{C_1}{\rho} Z(x) \frac{z}{z_0} A - 2C_2 \frac{B_1}{\rho r}. \quad (25)$$

Finalement

$$\frac{dV}{dt} = \pi r^2 \frac{d\ell}{dt} = \frac{1}{\rho} \frac{d(A + B_1)}{dt} = -2 \frac{C_2}{\rho} \frac{A + B_1}{\rho r},$$

ce qui donne la variation de  $\ell$

$$\frac{d\ell}{dt} = -2C_2 \frac{A + B_1}{\pi \rho^2 r^3} = -\frac{2C_2}{\rho r} \ell. \quad (26)$$

### 3.5 Modèle Hétérogène

On considère deux types de modèles hétérogènes pour tenir compte des caractéristiques de solubilité, de dégradabilité et d'absorbabilité. Le premier sera de type « phosphates », l'autre de type « amidon ». On n'a pas tenu compte des sécrétions mais il n'y a pas de difficulté à le faire.

#### 3.5.1 Le modèle type phosphates

Par rapport au modèle "cylindre" décrit par N. Traoré, on va effectuer un seul changement : l'unique nutriment  $A$  va être remplacé par trois types de nutriments aux propriétés différentes (notés  $A_1$ ,  $A_2$ ,  $A_3$ ). L'hétérogénéité viendra donc de la non-homogénéité de la partie nutriment mais pour le reste, on conserve une hypothèse d'homogénéité globale dans le cylindre : les concentrations en  $A_1$ ,  $A_2$ ,  $A_3$ ,  $B_1$  seront supposées être identiques en tout point du cylindre.

Les nutriments ont les propriétés suivantes :

- ◆  $A_1$  représente la masse des nutriments non solubles ; ils ne peuvent pas être attaqués par les enzymes, ils ne sont donc pas digérables en l'état. Mais leur état pourra évoluer car la proportion entre  $A_1$  et  $A_2$  fait l'objet d'une loi de comportement. *L'existence de cette loi de comportement est l'une des caractéristiques du modèle.*
- ◆  $A_2$  représente la masse des nutriments qui devront être d'abord dégradés en  $B_1$  (à l'intérieur du cylindre) avant d'être transformés, à la périphérie du cylindre, en  $B_2$  assimilable.
- ◆  $A_3$  représente la masse des nutriments qui sont transformables en  $B_2$  directement et ceci a lieu à la périphérie du cylindre (et uniquement là). On peut identifier  $A_3$  à  $B_1$  et c'est ce que nous ferons dans toute la suite ; la présence de  $A_3$  se traduira simplement par le fait que  $B_1(0)$  n'est pas nul.

On suppose :

- ① Les quantités  $A_1$  et  $A_2$  sont liés par une équations d'état  $A_1(t) = \mu(ph(x(t)))A_2(t)$  dépendant du  $ph$ . Donc, à  $ph$  fixé, la proportion de  $A_1$  est déterminée par le  $A_2$ .
- ② La dégradation de  $A_2$  se fait suivant la loi volumique mais il faut tenir compte de la loi liant  $A_1$  et  $A_2$  (l'analyse suivante est à affiner) : on note  $V(t)$  = volume du cylindre =  $(A_1 + A_2 + B_1)(t)$  (à constante près -masse volumique-) et  $[A_2] = A_2/V$  la concentration en  $A_2$ ; les variations de  $[A_2]$  ont trois origines : (i) la transformations de  $A_2$  en  $B_1$ , (ii) la variation de volume et (iii) la changement d'état de  $A_1$  en  $A_2$  (et inversement). Pour modéliser cette dernière transformation, on utilise le fait que la production ou consommation de  $A_1$  pour maintenir la loi " $\mu$ " se fait avec conservation de la masse : 1g de  $A_1$  produit 1g de  $A_2$ . La loi volumique devrait donner :

$$\frac{d[A_2]}{dt}(t) = -k_2(t).[A_2](t) - \frac{A_2(t)}{[V(t)]^2} \frac{dV}{dt}(t) - \frac{1}{V(t)} \frac{dA_1}{dt}(t),$$

et donc :

$$\frac{1}{V(t)} \frac{dA_2}{dt}(t) - \frac{A_2(t)}{[V(t)]^2} \frac{dV}{dt}(t) = -k_2(t) \cdot \frac{A_2(t)}{V(t)} - \frac{A_2(t)}{[V(t)]^2} \frac{dV}{dt}(t) - \frac{1}{V(t)} \frac{dA_1}{dt}(t),$$

ce qui conduit à :

$$\frac{dA_2}{dt}(t) = -k_2(t) \cdot A_2(t) - \frac{dA_1}{dt}(t),$$

On peut donc finir le calcul via :

$$\frac{dA_1}{dt}(t) = \mu \frac{dA_2}{dt}(t) + \frac{d\mu}{dt} A_2(t),$$

ce qui donne :

$$\frac{dA_2}{dt}(t) = \frac{1}{1 - \mu} \left[ -k_2(t) \cdot A_2(t) + \frac{d\mu}{dt} A_2(t) \right].$$

- ③ La dégradation de  $B_1$  se fait suivant la loi surfacique. L'équation sur  $B_2$  est donc la suivante (car la surface totale est du  $V^{1/2}$ ) :

$$\frac{dB_2}{dt}(t) = k_3(t) \cdot \frac{B_1(t)}{[V(t)]^{1/2}},$$

auquel il faut enlever la part assimilée. La variation de  $B_1$  s'obtient en combinant la production de  $B_1$  à partir du  $A_2$  et la dégradation de  $B_1$  en  $B_2$ , soit :

$$\frac{dB_1}{dt}(t) = k_2(t) \cdot A_2(t) - k_3(t) \cdot \frac{B_1(t)}{[V(t)]^{1/2}},$$

On sait calculer  $A_1$  (par la loi " $\mu$ "),  $A_2$  par son équation, de même que  $B_1$  et  $B_2$ . Reste  $V$  : la variation de volume est essentiellement celle de  $B_2$  donc :

$$\frac{dV}{dt}(t) = -k_3(t) \cdot \frac{B_1(t)}{[V(t)]^{1/2}}.$$

Le listing 11 montre une tentavie d'implantation du modèle, très simpliste car nous ne connaissons rien de la fonction  $\mu$  qui est prise constante (c.f. ligne 31). L'équation sur  $A1$  (ligne 21) à disparu car  $A1$  est calculé à partir de la relation  $A1 = \mu * A2$ .

---

```

1 //  MODELE TYPE "PHOSPHATES"
2 //  constante K  atténuation de la vitesse
3 //  constante c c0 c1 a b  voir modele vitesse
4 //  constante kb  disparition des enzymes
5 //  constante kt  absorption des nutriments.
6 //  constantes C1 pour dégradation A vers B1, et C2 pour A et B1 vers B2
7 //  x(1) = position
8 //  x(2) = dérivée position (ie vitesse)
9 //  x(3) : sans objet
10 //  x(4) = nutriments A2 dégradables : la proportion A1/A2 est une
11 //  fonction mu(x) du ph.
12 //  Le nutriment A3 directement assimilable devient la condition initiale
13 //  pour B1.
14 //  x(5) = digérés pas assimilables B1
15 //  x(6) = digérés assimilables B2
16 //  x(7) = rayon du cylindre
17 //  x(8) = enzymes
18 function f=digPhos(t,x)
19     f(1)= x(2);
20     f(2)= - K*x(2) + W(t-x(1)/c)*(c0+c1*x(4)*mu(x(1)))/(a+b*x(1))*(1-x(2)/c) ;
21     f(3) = 0;
22     f(4) = (C1*ph(x(1))*x(8)/z0/rho+ dmu(x(1))*x(2)*x(4))/(1-mu(x(1)));
23     f(5)= - C2*2*sqrt(Pi*l/rho)*x(5)/sqrt(abs(x(4)*(1+mu(x(1))+x(5))))..
24         + C1*ph(x(1))*x(8)*x(4)*mu(x(1))/z0/rho;
25     f(6)= C2*2*sqrt(Pi*l/rho)*x(5)*sqrt(abs(x(4)*(1+mu(x(1))+x(5)))) - kt*x(6);
26     f(7)= - C2/rho;
27     f(8)= - kb*x(8);
28 endfunction
29
30 //  equilibre entre A1 et A2
31 function y = mu(x)
32     y = 0.5;
33 endfunction
34
35 //  dérivée de la précédente
36 function y = dmu(x)
37     y = 0.0;
38 endfunction

```

---

Le programme d'exécution est quasiment identique au précédent (listing 7). Un exemple d'exécution est donné par la figure 9.

### 3.5.2 Le modèle type amidon

Par rapport au modèle de base, l'unique nutriment  $A$  va être remplacé par *deux* types de nutriments aux propriétés différentes (notés  $A_1$ ,  $A_2$ ); typiquement  $A_1$  désigne la partie « amidon » et  $A_2$  la partie « glucose » qui en résulte. On ne fait pas de différence ici non plus entre  $A_3$  et  $B_1$  : la présence de glucose facilement assimilable à l'entrée dans l'intestin grêle sera représenté par une donnée initiale non nulle de  $B_1$ .




FIG. 9 – Modèle de type « phosphate »

L'hétérogénéité sera plus marquée dans ce second modèle puisque nous supposons que  $A_1$  occupe lui-même un cylindre, placé au centre du cylindre habituel. Ce cylindre central représente donc une « zone morte », qui sera tout de même attaquée en surface par des enzymes pour que  $A_1$  soit transformé en  $A_2$ . Dans le reste du cylindre, on conserve une hypothèse d'homogénéité en  $A_2$ - $B_1$  : les concentrations en  $A_2$  et  $B_1$  seront supposées être identiques en tout point du cylindre, hors la zone morte.

On suppose :

- ♦ La dégradation de  $A_1$  en  $A_2$  se fait via une enzyme (notée  $z_1$ ) et de manière surfacique au bord du cylindre  $A_1$ , soit :

$$\frac{dA_1}{dt}(t) = -C_1 k_1(ph(x(t))).z_1^{1/2}(t)A_1^{1/2}(t) .$$

Le terme en  $z_1^{1/2}$  s'interprète comme une *concentration surfacique* en enzyme. On rappelle que, sur  $z_1$ , on peut écrire une équation du type :

$$\frac{dz_1}{dt}(t) = -\bar{k}_1 z_1(t) .$$

- ♦ La dégradation de  $A_2$  se fait suivant la loi volumique en utilisant une enzyme  $z_2$  et on devrait avoir :


$$\frac{dA_2}{dt}(t) = C_1 k_1(ph(x(t)), z_1(t)).A_1^{1/2}(t) - C_2 k_2(ph(x(t)), z_2(t)).A_2(t) ,$$

le premier terme venant de la production de  $A_2$  par  $A_1$  et le second de la dégradation volumique. On a sur  $z_2$  une équation de type  $z_1$ .


- ♦ La transformation finale de  $B_1$  en  $B_2$  se fait surfaciquement et donne donc :

$$\frac{dB_2}{dt}(t) = +k_3(t).(A_1(t) + A_2(t) + B_1(t))^{1/2} \frac{B_1(t)}{A_2(t) + B_1(t)} ,$$

le second membre étant donc proportionnel à la surface de  $B_1$  à la périphérie du cylindre (il faudra enlever l'absorption par les parois de l'intestin en  $-\tilde{k}B_2$ ).



(a) Vitesse et Positions



(b) Nutriments et Digérés

FIG. 10 – Modèle mixte

- ◆ Reste l'équation sur  $B_1$  qui s'obtient en agrégeant celles sur la dégradation de  $A_2$  en  $B_1$  et celle de  $B_1$  en  $B_2$  :

$$\frac{dB_1}{dt}(t) = -k_3(t) \cdot (A_1(t) + A_2(t) + B_1(t))^{1/2} \frac{B_1(t)}{A_2(t) + B_1(t)} + C_2 k_2(ph(x(t)), z_2(t)) \cdot A_2(t) ,$$

### 3.6 Modèle à deux vitesses

On pourrait envisager, à partir des modèles précédents, une évolution en deux temps. Le premier temps correspondant au modèle impulsion, avec des valeurs de paramètres permettant de parcourir une petite portion de l'intestin dans un temps de 3 ou 4 heures, que l'on pourrait assimiler à la *phase II* (c.f Appendice) de la digestion. Le deuxième temps correspondrait à une évolution assurée par les *complexes myoélectriques migrants* (c.f Appendice) jouant le rôle de « voiture balai » (*housekeeper*) qui propagent le paquet sans à-coups. On peut réaliser ceci en modifiant les premières lignes de la fonction *digABB* de la façon suivante

Listing 12 – Fonction digCMM

---

```

1 function f=digCMM(t , x)
2   f(1)= c*(c0+c1*x(3))/(a+b*x(1))*bool2s(t>=transit)+x(2)*bool2s(t<transit)
3   f(2)= bool2s(t<transit)*(-K*x(2) + W(t-x(1)/c)*(c0+c1*x(3))/(a+b*x(1))*(1-x(2)/c)
4   )
5   .....
6 endfunction

```

---

Ce qui donne les résultats de la figure 10.

## 4 Perspectives

Les travaux présentés ci-dessus constituent une première étape dans la modélisation de la digestion dans l'intestin grêle avec deux axes principaux : l'un consacré aux problèmes de transport du *bolus*, c'est-à-dire du ou des « paquets », et l'autre à la dégradation de ce bolus par les enzymes (de manière volumique et surfacique).

La modélisation de ces deux phénomènes devra être affinée, ce qui donnera lieu à deux directions de recherche à la fois indépendantes et complémentaires.

Bien entendu, un élément essentiel dans la démarche de modélisation sera l'étape de validation, au moins partielle, de nos modèles par les physiologistes qui passera inévitablement par l'acquisition de données quantitatives. Il sera aussi nécessaire de faire des analyses de sensibilité, pour chacun des modèles et pour divers paramètres, et de les comparer si possible avec des données numériques afin de calibrer les valeurs.

Très récemment, nous avons introduit un autre type de modèle, dit « à compartiments » basé sur un automate cellulaire (cellule étant pris ici au sens informatique et non au sens biologique) qui donne des résultats encourageants pour ce qui concerne la dégradation des nutriments mais devra être repensé au niveau du transport.

Enfin une thèse cofinancée par l'INRA et le CNRS permettra de développer ces différentes perspectives.

# Annexes

## A Motricité intestinale

La paroi de l'intestin grêle est formée de deux couches de muscles lisses dont l'activité doit être coordonnée :

- ♦ une couche longitudinale externe,
- ♦ une couche circulaire interne.

Les cellules de Cajal (CIC), situées entre les deux couches musculaires, sont à l'origine de cet automatisme. Les fibres se dépolarisent régulièrement et donnent naissance à des *ondes lentes*. Ces ondes lentes ne sont pas elles mêmes à l'origine d'une activité mécanique : il faut qu'il y ait en plus un potentiel d'action (pointes ou *spikes*). La fréquence des ondes lentes décroît dans le sens oral-aboral (18 par minute au niveau du duodénum, 12 par minute au niveau de l'iléon chez l'homme) de façon discontinue sur chaque segment d'intestin grêle [Phd10].

Le principal rôle du système nerveux entérique est d'assurer la coordination des deux couches musculaires dans le sens oral-aboral lors de la contraction intestinale ; l'exemple prototype est le réflexe péristaltique.

**Loi de l'intestin (1899)** Lorsque l'intestin est soumis à une excitation (mécanique, chimique ou électrique) on observe une contraction en amont sur 2-3 cm et une inhibition en aval sur 6-7 cm [Phd10].

Pour un segment donné de l'intestin grêle, et selon la fréquence des pointes on va distinguer trois niveaux d'activité [Phd11] :

- ♦ Phase I (le repos) : repos de l'intestin
- ♦ Phase II (activité irrégulière) : mélange du contenu favorisant l'absorption (segmentation et mouvement pendulaire)
- ♦ Phase III (activité régulière) : activité de balayage (housekeeper) pour faire transiter les résidus (débris cellulaires, bactéries,...) vers l'aval. Toutes les ondes lentes sont suivies de potentiels de pointe et sont associées à une contraction intestinale.




FIG. 11 – Les mouvements de l'intestin grêle [CH5S2]. Grâce à la contraction longitudinale, le bol avance dans l'intestin comme le pied dans une chaussette [Phd10]

Sur un même segment de l'intestin ces trois types d'activité se succèdent au cours du temps pour former un complexe moteur migrant CMM (ou encore complexe myoélectrique migrant) qui se déplace progressivement tout au long de l'intestin.

Pour un chien *à jeun* depuis 10 heures on peut observer que la phase I dure environ 20mn, le phase II qui suit dure de 40 à 60 minutes et se termine par une phase III de durant 5 à 7 minutes. Sur la Figure 12 les 7 bandes sont consécutives dans le temps et illustrent le déroulement complet d'un complexe myoélectrique en un même site intestinal [LA-RO] :

- ◆ En 1, une phase d'activité de pointes régulières (phase III) dure 4 mn 30.
- ◆ Elle est suivie, après une période intermédiaire (pointillés) par une période de quiescence (phase I) durant un peu plus de 5 mn.
- ◆ Puis on observe le déroulement jusqu'au début du tracé 7, de la phase II d'activité de pointes irrégulières.
- ◆ Enfin est enregistré une nouvelle phase III (4 mn) suivie de quiescence.

La séquence complète du début de la phase I à la fin de la phase III dure 70 mn.

Chez toutes les espèces la durée de propagation est de 90 à 120 minutes, donc plus rapide pour le cheval (longueur de l'intestin grêle 30m) que pour le chien (longueur 3m).

- ◆ Porc : 120 minutes
- ◆ Chien : 97 à 120 minutes
- ◆ Lapin : 86 à 107 minutes
- ◆ Mouton : 103 à 110 minutes

Lorsque la phase III atteint l'iléon, une autre prend naissance au niveau duodénal. Pour l'homme (et le chien) il existe au moins un et au plus deux MMC sur l'intestin grêle à un instant donné ; pour le porc et le veau ce sont deux ou trois MMC qui sont présents [LA-RO].

Les CMM démarrent en fait dans l'estomac et sont coordonnés avec la vidange biliaire et pancréatique : un bolus de bile est délivré dans le duodénum pendant la phase II. La vidange se termine avec la phase III qui va pousser la bile et les acides biliaries vers l'iléon.

Tout ce qui précède est valable pour le sujet **à jeun**.

La prise de nourriture apporte des modifications importantes chez les monogastriques : il y a suppression des CMM pendant une durée de 8 à 12 heures selon la nature du repas. On retrouve à la place une activité de type II sur l'ensemble de l'intestin grêle.[Phd11]

L'influence de la prise de nourriture sur les CMM est très différente selon les espèces :




FIG. 12 – Electromyogramme duodénal global, dérivé à 16 cm du pylore chez le porc et enregistré en méthode bipolaire et faible constante de temps (0,03s).




FIG. 13 – La superposition des valvules, villosités et microvillosités multiplie la surface d'absorption par 600 pour aboutir à une surface totale d'échanges de  $200\text{ m}^2$ .

- ◆ chez les ruminants la prise de nourriture est sans effet,
- ◆ chez le cheval la suppression ne dure que 1 à 2 heures,
- ◆ chez le porc la suppression dure 3 heures si l'animal est nourri 2 fois par jour et 6 heures s'il ne reçoit qu'un seul repas.

Ces différences sont dues à l'apport plus ou moins continu du chyme par l'estomac, continu chez les bovins et discontinu chez les carnivores.

Chez les monogastriques comme le chien le débit duodénal est de 430 ml/h, une heure après le repas et de 125 ml/h pour l'heure suivante. La progression du chyme se fait avec les ondes péristaltiques propagées et globalement cela est plus lent (-80%) que chez l'animal à jeun. Cela veut dire que les très nombreuses contractions vues pendant la phase post-prandiales jouent plutôt un rôle de frein et facilitent les mélanges et la digestion [Phd11].

La grande capacité d'absorption de l'intestin grêle est liée à sa structure que l'on peut représenter comme une superposition de plis de plus en plus petits. De ce fait la surface d'absorption réelle est fortement augmentée. On décrit ainsi [DIG2] (cf Figure 13)

- ◆ Les valvules conniventes qui sont des replis transversaux de la muqueuse intestinale de 1 à 2 cm de hauteur.
- ◆ Les villosités intestinales qui sont des plis muqueux de 1 mm environ. Ces villosités sont séparées par des cryptes qui constituent la zone de germination des cellules intestinales.
- ◆ Les microvillosités intestinales qui sont situées au pôle apical de l'entérocyte et constituent la bordure en brosse. Ces microvillosités sont recouvertes d'un matériau fibrillaire riche en glycoprotéines, le *glycocalyx* qui joue un rôle de filtre du contenu intestinal, participe à la constitution d'une couche aqueuse non agitée et dont certaines de ses fibres font partie des molécules d'enzymes digestives ancrées dans la membrane des microvillosités.

## B Hydrolyse Enzymatique et Digestion

Les fonctions du système digestif sont

- ◆ La dégradation de la nourriture en plus petites molécules





FIG. 14 – Constituants de l'amidon : (a) amylose, (b) amylopectine.

- ◆ L'absorption de ces petites molécules puis leur transport.

La digestion est une réaction d'hydrolyse qui brise les liaisons des macromolécules : à partir d'un molécule d'eau  $H-OH$  elle brise une liaison pour rajouter  $H$  à une des molécules et  $OH$  à l'autre.



L'hydrolyse est une réaction spontanée très lente qui est accélérée par des enzymes. Une hydrolyse enzymatique demande moins d'énergie (temps et chaleur), est beaucoup plus rapide et nécessite des conditions opératoires moins extrêmes (moins dangereuses) qu'une hydrolyse chimique (i.e acide) : c'est donc un système beaucoup plus efficace. Les enzymes n'affectent que la vitesse de la réaction et jamais son équilibre. Une réaction défavorisée reste défavorisée même en présence d'enzyme.

Chaque enzyme est spécifique du produit (substrat = substance sur laquelle agit une enzyme) qu'elle hydrolyse (*spécificité de substrat*)

- ◆ les protéases dégradent les protéines
- ◆ les lipases dégradent les lipides
- ◆ les nucléases dégradent les acides nucléiques
- ◆ les amylases, maltases, etc ... dégradent les hydrates de carbone (glucides).

## B.1 Glucides

Chez les monogastriques, les glucides (amidon et cellulose) représentent une part importante de l'apport énergétique et les produits terminaux de la digestion sont des monosaccharides (glucose, galactose, fructose) qui sont les molécules absorbables par le tube digestif.

L'amidon est un polysaccharide  $(C_6H_{10}O_5)_n$  avec des liaisons de type  $\alpha$ -(1-4) glucose ou  $\alpha$ -(1-6) glucose. La cellulose est un polymère formé par des liaisons de type  $\beta$ -(1-4) glucose, c'est une molécule linéaire. Une fibre de cellulose est formée par environ 500 000 molécules.

Les mammifères ne possèdent pas d'enzymes capables d'attaquer les liaisons  $\beta$ -(1-4) glucose : la digestion de la cellulose se fera obligatoirement par fermentation microbienne, les bactéries apportant leur faculté enzymatique pour hydrolyser les glucides.

Chez l'homme, la réaction d'hydrolyse de l'amidon qui correspond en fait à la digestion de l'amidon n'est pas catalysée par de l'acide chlorhydrique mais par des enzymes (catalyseurs produits par des organismes vivants) : on parle d'hydrolyse enzymatique. Cette réaction commence dans la bouche et se termine dans l'intestin grêle





FIG. 15 – Endo et exopeptidases [Phd13]



Une amylase est une saccharidase, c'est-à-dire un enzyme clivant les polysaccharides. L' $\alpha$ -amylase pancréatique attaque les liaisons  $\alpha$ -(1-4) terminales et les  $\alpha$ -(1-4) voisines des liaisons  $\alpha$ -(1-6). Cela transforme les polysaccharides en oligo-saccharides (3 à 10 monosaccharides) ; les oligo-saccharides sont hydrolysés par les enzymes *brush-borders* ce qui conduit à la production de maltose (2 glucoses) et de maltotriose (3 glucoses) Seuls les monosaccharides sont absorbés par la bordure en brosse.

Rappel :

- ◆ Le glucose forme un précipité rouge brique en présence de liqueur de Fehling et à chaud ; ce test met également en évidence l'isomère du glucose qui est le fructose.
- ◆ L'amidon prend une coloration bleue-noire en présence d'eau iodée.

## B.2 Protéines

C'est grosso-modo le même mécanisme pour les protéines : les protéases assurent la digestion des protéines, mais interviennent de façon séquentielle. On peut distinguer :

- ◆ les endopeptidases (pepsine, trypsine, chymotrypsine, chymosine<sup>1</sup>) qui attaquent les liaisons peptidiques situées à l'intérieur des protéines
- ◆ les exopeptidases (carboxypeptidase luminale et aminopeptidase de la bordure en brosse) qui attaquent les liaisons peptidiques possédant un radical  $COOH$  ou  $NH_2$  libre.

La pepsine est libérée au niveau de l'estomac et nécessite un milieu acide (HCl avec un pH compris entre 1 et 2). Pour que la trypsine agisse de façon optimale, il faut que les protéines aient subi l'action de la pepsine et qu'il y ait présence d'ion  $Ca^{++}$  avec un milieu alcalin. La trypsine déclenche la sécrétion des autres endopeptidases et des exopeptidases.

<sup>1</sup>encore appelée rennine ou présure

## B.3 Lipides

Les graisses sont un problème pour la digestion puisqu'elles sont hydrophobes : elles s'associent entre elles pour former de gros globules. Ensuite :

- ◆ Les gros globules sont émulsifiés par les sels biliaires dans le duodénum : production de gouttelettes de graisse recouvertes de sels biliaires.
- ◆ La digestion des graisses par la lipase, une enzyme pancréatique, donne des acides gras libres et des MAG (monoacylglycerols) qui forment des micelles.
- ◆ Les acides gras et MAG quittent les micelles et entrent dans les cellules épithéliales par diffusion. item Les graisses sont recouvertes de protéines pour former les chylomicrons qui sont transportés hors de l'intestin. Ils sont trop gros pour passer directement dans les capillaires sanguins mais peuvent entrer dans le capillaire lymphatique (chylifère) et finissent par rejoindre la circulation sanguine.

## Références

- [ANAPH] <http://www.ustboniface.mb.ca/cusb/abernier/Biologie/Anat-physio/digestionCH41-08.pdf>
- [Phd10] [http://physiologie.envt.fr/spip/IMG/pdf/Phys\\_digest\\_10.pdf](http://physiologie.envt.fr/spip/IMG/pdf/Phys_digest_10.pdf)
- [Phd11] [http://physiologie.envt.fr/spip/IMG/pdf/Phys\\_digest\\_11.pdf](http://physiologie.envt.fr/spip/IMG/pdf/Phys_digest_11.pdf)
- [Phd13] [http://physiologie.envt.fr/spip/IMG/pdf/Phys\\_digest\\_13.pdf](http://physiologie.envt.fr/spip/IMG/pdf/Phys_digest_13.pdf)
- [CAG5] [http://www.cag-acg.org/uploads/firstprinciples/z\\_pdf/FCH\\_5\\_06.pdf](http://www.cag-acg.org/uploads/firstprinciples/z_pdf/FCH_5_06.pdf)
- [DIG2] <http://www.carabin.fr/~telech//cours/PCEM%202/Physiologie/digestion2.pdf>
- [CH5S2] <http://coproweb.free.fr/pagphy/physioan/ch5s2.htm#top>
- [CH5S2] <http://coproweb.free.fr/pagphy/physioan/ch5s4.htm#top>
- [LA-RO] J.P. LAPLACE, C. ROMAN **Activités de la musculature gastro-intestinale et mouvements des contenus digestifs** *Ann. Biol. anim. Bioch. Biophys.*, 1979, **19**(3 B), 849-879.
- [CH-PH] P. ATKINS, J. DE PAULA *Chimie physique pour les sciences de la vie* DUNOD (2005)